/*-----------------------------------------------------------------------------
	The contents of this file are subject to the Mozilla Public License
	Version 1.1 (the "License"); you may not use this file except in
	compliance with the License. You may obtain a copy of the License at
	http://www.mozilla.org/MPL/
	
	Software distributed under the License is distributed on an "AS IS"
	basis, WITHOUT WARRANTY OF ANY KIND, either express or implied. See the
	License for the specific language governing rights and limitations
	under the License.
	
	The Original Code is Fever Framework code.
	
	The Initial Developer of the Original Code is Romain Ecarnot.
	Portions created by Initial Developer are Copyright (C) 2006
	the Initial Developer. All Rights Reserved.
	
	Contributor(s):
	
	Unless required by applicable law or agreed to in writing, software
	distributed under the License is distributed on an "AS IS" BASIS,
	WITHOUT WARRANTIES OR CONDITIONS OF ANY KIND, either express or implied.
	See the License for the specific language governing permissions and
	limitations under the License.
-----------------------------------------------------------------------------*/

import fever.exception.IllegalArgumentException;
import fever.utils.Stringifier;
import fever.utils.StringUtil;

/**
 * Locked day definition structure.
 * 
 * @author Romain Ecarnot
 */
class fvaswing.components.datechooser.FvLockDay
{
	//-------------------------------------------------------------------------
	// Private properties
	//-------------------------------------------------------------------------
	
	private static var _SEPARATOR : String = ':';
	
	private var _day : Number;
	private var _month : Number;
	
		
	//-------------------------------------------------------------------------
	// Public API
	//-------------------------------------------------------------------------
	
	/**
	 * Parses passed-in <code>list</code> array containing day definition 
	 * in string format, into array with <code>LockDay</code> structure.
	 */
	public static function parseList( list : Array ) : Array
	{
		var l : Number = list.length;
		var a : Array = new Array();
		
		for ( var i : Number = 0; i < l; i++ ) 
		{
			a.push( new FvLockDay( list[ i ] ) );
		}
		
		return a;
	}
	
	/**
	 * Returns <code>true</code> if passed-in <code>m</code> (month) and 
	 * <code>y</code> (year) are equal to current month and year values.
	 * 
	 * @return Boolean
	 */
	public function isEqualDayAndMonth( d : Number, m : Number ) : Boolean
	{
		return ( m == _month && d == _day ); 	
	}
	
	/**
	 * Returns string representation.
	 */
	public function toString() : String
	{
		return Stringifier.parse( this, 'Lock day :: ' + _day + '/' + _month );
	}
	
	
	//-------------------------------------------------------------------------
	// Private implementation
	//-------------------------------------------------------------------------
	
	private function FvLockDay( definition : String ) 
	{
		if( !StringUtil.contains( definition, _SEPARATOR ) )
		{
			throw new IllegalArgumentException( 'Locked day definition is malformed :: ' + definition );	
		}
		else
		{
			var a : Array = definition.split( _SEPARATOR );
			
			_day = Number( a[0] );
			_month = Number( a[1] );
		}
	}	
}